NAG Library Routine Document

F07CGF (DGTCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CGF (DGTCON) estimates the reciprocal condition number of a real n by n tridiagonal matrix A, using the LU factorization returned by F07CDF (DGTTRF).

2 Specification

```
SUBROUTINE F07CGF(NORM, N, DL, D, DU, DU2, IPIV, ANORM, RCOND, WORK,
1                 IWORK, INFO)
INTEGER           N, IPIV(*), IWORK(N), INFO
double precision  DL(*), D(*), DU(*), DU2(*), ANORM, RCOND, WORK(2*N)
CHARACTER*1       NORM
```

The routine may be called by its LAPACK name ***dgtcon***.

3 Description

F07CGF (DGTCON) should be preceded by a call to F07CDF (DGTTRF), which uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one nonzero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals. F07CGF (DGTCON) then utilizes the factorization to estimate either $\left\|A^{-1}\right\|_1$ or $\left\|A^{-1}\right\|_\infty$, from which the estimate of the reciprocal of the condition number of A, $1/\kappa(A)$ is computed as either

$$1/\kappa_1(A) = 1/\left(\|A\|_1\left\|A^{-1}\right\|_1\right)$$

or

$$1/\kappa_\infty(A) = 1/\left(\|A\|_\infty\left\|A^{-1}\right\|_\infty\right).$$

$1/\kappa(A)$ is returned, rather than $\kappa(A)$, since when A is singular $\kappa(A)$ is infinite.

Note that $\kappa_\infty(A) = \kappa_1\left(A^{\mathrm{T}}\right)$.

4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

1: NORM – CHARACTER*1 *Input*

On entry: specifies the norm to be used to estimate $\kappa(A)$.

NORM = '1' or 'O'
Estimate $\kappa_1(A)$.

NORM = 'I'
Estimate $\kappa_\infty(A)$.

Constraint: NORM = '1', 'O' or 'I'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: DL($*$) – ***double precision*** array *Input*

Note: the dimension of the array DL must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n-1)$ multipliers that define the matrix L of the LU factorization of A.

4: D($*$) – ***double precision*** array *Input*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the upper triangular matrix U from the LU factorization of A.

5: DU($*$) – ***double precision*** array *Input*

Note: the dimension of the array DU must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n-1)$ elements of the first superdiagonal of U.

6: DU2($*$) – ***double precision*** array *Input*

Note: the dimension of the array DU2 must be at least $\max(1, \mathrm{N} - 2)$.

On entry: must contain the $(n-2)$ elements of the second superdiagonal of U.

7: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n pivot indices that define the permutation matrix P. At the ith step, row i of the matrix was interchanged with row IPIV(i), and IPIV(i) must always be either i or $(i+1)$, IPIV(i) $= i$ indicating that a row interchange was not performed.

8: ANORM – ***double precision*** *Input*

On entry: if NORM = '1' or 'O', the 1-norm of the **original** matrix A.

If NORM = 'I', the ∞-norm of the **original** matrix A.

ANORM may be computed by calling F06RNFwith the same value for the parameter NORM.

ANORM must be computed either **before** calling F07CDF (DGTTRF) or else from a **copy** of the original matrix A (see Section 9).

Constraint: ANORM ≥ 0.0.

9: RCOND – ***double precision*** *Output*

On exit: contains an estimate of the reciprocal condition number.

10: WORK($2 \times$ N) – ***double precision*** array *Workspace*

11: IWORK(N) – INTEGER array *Workspace*

12: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

8 Further Comments

The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization. The total number of floating-point operations required to perform a solve is proportional to n.

The complex analogue of this routine is F07CUF (ZGTCON).

9 Example

This example estimates the condition number in the 1-norm of the tridiagonal matrix A given by

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix}.$$

9.1 Program Text

```
*     F07CGF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=50)
*     .. Local Scalars ..
      DOUBLE PRECISION ANORM, RCOND
      INTEGER          I, INFO, N
*     .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), DL(NMAX-1), DU(NMAX-1), DU2(NMAX-2),
     +                 WORK(2*NMAX)
      INTEGER          IPIV(NMAX), IWORK(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION F06RNF, X02AJF
      EXTERNAL         F06RNF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         DGTCON, DGTTRF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07CGF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the tridiagonal matrix A from data file
*
         READ (NIN,*) (DU(I),I=1,N-1)
         READ (NIN,*) (D(I),I=1,N)
```

```
         READ (NIN,*) (DL(I),I=1,N-1)
*
*        Compute the 1-norm of A
*
         ANORM = F06RNF('1-norm',N,DL,D,DU)
*
*        Factorize the tridiagonal matrix A
*
         CALL DGTTRF(N,DL,D,DU,DU2,IPIV,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Estimate the condition number of A
*
            CALL DGTCON('1-norm',N,DL,D,DU,DU2,IPIV,ANORM,RCOND,WORK,
     +                  IWORK,INFO)
*
*           Print the estimated condition number
*
            IF (RCOND.GE.X02AJF()) THEN
               WRITE (NOUT,99999) 'Estimate of condition number = ',
     +           1.0D0/RCOND
            ELSE
               WRITE (NOUT,99999)
     +           'A is singular to working precision. RCOND = ', RCOND
            END IF
*
         ELSE
            WRITE (NOUT,99998) 'The (', INFO, ',', INFO, ')',
     +        ' element of the factor U is zero'
         END IF
       ELSE
          WRITE (NOUT,*) 'NMAX too small'
       END IF
*
99999 FORMAT (1X,A,1P,E10.2)
99998 FORMAT (1X,A,I3,A,I3,A,A)
      END
```

9.2 Program Data

```
F07CGF Example Program Data
   5                          :Value of N
        2.1  -1.0   1.9   8.0
  3.0   2.3  -5.0  -0.9   7.1
  3.4   3.6   7.0  -6.0        :End of matrix A
```

9.3 Program Results

```
 F07CGF Example Program Results

 Estimate of condition number =   9.27E+01
```